
Introduction
Campbell Scientific dataloggers can send emails using EmailSend(), which allows the datalogger to issue notifications about 
the status of a station, transfer data files as attachments, or alert someone about an alarm condition. 

EmailSend() requires the use of an SMTP server. Using publicly available free services such as Gmail or Yahoo, introduce inher-
ent problems that can restrict the datalogger from sending emails. Campbell Scientific has no control over the development 
of these services, which becomes a moving target. None of these services are designed for the needs and limitations of a low-
power embedded device. Their requirements are much better suited for a PC than a datalogger. 

With EmailRelay(), the datalogger uses HTTPS instead of SMTP. The datalogger is basically posting a message to the service 
saying, “take this information and pass it along as an email.”  This eliminates the barriers that are common with these free ser-
vices because we now control the connection and authentication process from the datalogger to the cloud. This also allows us 
to easily adapt to changes in email services without changing settings in your datalogger. 

EmailRelay() Description
EmailRelay() gives the datalogger the ability to issue notifications about the status of the datalogger, transfer data files as 
attachments, and alert someone about an alarm condition, all via email. To successfully use EmailRelay(), the datalogger must 
have at least outbound Internet connectivity and be configured to use DNS. 

EmailRelay() was introduced on the following operating systems:

OS 31 for the CR1000, CR3000, and CR800-series dataloggers
OS 6 for the CR6 datalogger
OS 5 for the CR300-series dataloggers

EmailRelay() has four required parameters:

 EmailRelay(ToAddr,Subject,Message,ServerResponse)

The ToAddress (ToAddr) parameter is the email address you would like the message sent to. The FromAddress of the email sent 
by the datalogger is emailrelay@konectgds.com. This address reflects the server that is hosting Campbell Scientific’s cloud-
hosted data collection platform, Konect GDS. 

EmailRelay() also has optional parameters that let you send attachments or send data directly from a data table, without first 
writing the data to a file (referred to as streaming). 

With EmailRelay(), you can send up to 100 emails each day, and each message can be up to 1 MB in size, which includes the 
message itself, the message overhead, and any attachments. Sending the message to multiple addresses does not increase the 
number of email counts or the total size of the message. The message count is reset each day. 

Example Programs
Below are example programs that send an email message when a low battery voltage occurs. The email@provider.com address 
used in the constant ToAddr needs to be changed to the actual email address where the message will be sent. 

Example 1—Program that sends alarm on low-battery voltage measurement
'declare program variables and constants
Public batt_volt
Public alarm_sent As Boolean
Public email_result As String * 255

'main program
BeginProg
   alarm_sent = 0
   Scan (1,Sec,30,0)
      Battery (batt_volt)
   NextScan
   SlowSequence)
     Scan (10,Sec,0,0)
      If batt_volt < 10.5 Then
         If  alarm_sent = 0 Then
           alarm_sent = EmailRelay("email@provider.com", "Low Battery Warning", "Battery Voltage: " & batt_volt & "v", email_result)
         EndIf
      Else
         alarm_sent = 0
         Erase(email_result)
      EndIf
   NextScan
EndProg

Example 2—Test Program that sets the alarm trigger manually 
'declare program variables and constants
Const ToAddr = "email@provider.com"
Const Subject = "Email Message Test"
Const CRLF =  CHR(13) & CHR(10)

Public Batt
Public AlarmTrigger As Boolean
Public Message As String * 250
Public EmailSuccess
Public ServerResponse As String * 50

'main program
BeginProg
   Scan (1,Sec,3,0)
      Battery (Batt)
   NextScan
   SlowSequence
     Scan (1,Sec,1,0)

  'set alarm trigger manually
         If  AlarmTrigger Then
           Message = "Hello!" & CRLF & CRLF
           Message = Message & "This is an automatic email message from our friendly datalogger named   " & Status.StationName & "." 
           Message =  Message & "An alarm condition has been triggered."
           Message =  Message & "The battery voltage is" & Batt & "volts" & CRLF & CRLF & CRLF
            Message =  Message & "Datalogger time is" & Status.Timestamp
           EmailSuccess =  EmailRelay(ToAddr,Subject,Message,ServerResponse)
           AlarmTrigger  =  False
         EndIf
         Erase(Message)    'Erase the message after sending
   NextScan
EndProg

Helpful Tips
When creating and testing a program using EmailRelay(), use the manual trigger to run EmailRelay() and set the trigger back 
to False after its execution (Example 2). This ensures you won’t exceed the daily email limit during the first test run of your pro-
gram. Run EmailRelay() in a SlowSequence to avoid the delay of critical measurements and other important tasks executed 
in the main scan. 


